
//***************************************************************************** 
// PROGRAMME  : TDGACORUS 2010.pgm (pgm Epidata Analysis) 
// DATE       : début le 29/03/2010 
// AUTEUR     : P. Traissac, IRD, UMR 204 NUTRIPASS, pierre.traissac@ird.fr 
// LIEU       : Bureau PT IRD Montpellier 
// BUT        : Programme récapitulatif de tous les exercices de gestion 
//              Ecole Thématique CORUS 2010 Rabat 
// 
// DONNEES EN ENTREE :- femmes2010.rec 
//                      u.s. = femme (n=1849) 
//                     - menages2010.rec 
//                      u.s. = femme (n=1763) 
// 
//    EN SORTIE : - femmes2010_2.rec  données femmes avec nouvelles variables 
//                u.s. = femme (n=1849) 
//                - menages2010_2.rec  données menages avec nouvelles variables 
//                       u.s. = menage (n=1763) 
//                - menfem2010_2.rec données femmes + ménages 
//                         avec nouvelles variables 
//                       u.s. = femme (n=1849) 
//***************************************************************************** 
// Répertoir des données (à modifier suivant le cas) 
cd "D:\corus\maroc\FormationRabat2010\data" 
 
 
//****************************************** 
//             Exercice 1 
//             Manipulations élémentaires 
//****************************************** 
 
// *********** Données ménages************ 
 
read "menages2010.rec" /close 
// vérification nombre d'observation et minimum maximum 
describe 
// Visualisation de la table 
browse 
// afficher seulement certaines variables 
browse idmen region typeloge 
//list   idmen region typeloge 
// par depenses croissantes 
sort depenses 
browse idmen depenses 
// par habitat et eco1 
sort habitat eco1 
// modalités des variables 
freq habitat typeloge mode 
// en faisant apparaitre  "donnée manquante" 
freq habitat typeloge mode /m 

//****************************************** 
//             Exercice 1 
//             Manipulations élémentaires 
//****************************************** 
 
// *********** Données femmes ************ 
 
read "femmes2010.rec" /close 
// vérification nombre d'observation et minimum maximum 
describe 
// modalités des variables 
freq actpc2 matc3 
// en faisant apparaitre  "donnée manquante" 
freq actpc2 matc3 /m 
// sélection des femmes 
select if cal>=2500 
// nbre de femmes sélectionnées 
count 
//et vérification de la sélection 
describe 
// ou bien 
sort cal 
browse idfem cal 
// retour à l'ensemble des femmes 
select 
// vérification 
count 
// sélection des femmes 
select if cal < 2500 and actpc2 = 1 
// nbre de femmes 
count 
// vérification 
freq actpc2 
des cal 
 
// retour sélection générale 
select 
// creation table femmemarie.rec 
// selection 
select if matc3=2 
//sauvegarde 
savedata "femmemarie.rec" idfem anais jnais mnais poids taille tt th anais 
jnais mnais /replace 
 
// on ouvre la table 
read  "femmemarie.rec" /close 
// nbre de femmes nées avant 1960 dans cette table 
select if anais>1960 
count 
// nbre de femmes nées avant 1960 sur la table initiale 
read "femmes2010.rec" /close 
select if anais>1960 
count 
// on supprime la table femmemarie.rec 
erase "femmemarie.rec" 

//************************************************** 
//         Exercice 2 
//         Concaténation de tables 
//************************************************** 
 
read "femmes2010.rec" /close 
merge idmen /file="menages2010.rec" /table 
// distribution de deux variables "femme" 
// % idem sur tableau des femmes 
freq actpc2 matc3 /m /c 
// distribution de deux variables "ménage" 
// %  ne sont pas les bons car certains ménages dupliqués 
freq typeloge mode /m /c 
 
// Distribution de mode d'occupation 
// Femmes avec activité professionnelle 
freq mode /c /m  if actpc2=1 
// Femmes sans activité professionnelle 
freq mode /c /m  if actpc2=2 
// autre façon, avec tableau croisé 
tables mode actpc2 /m /r 
 
//sauvegarde de la table fusionnée 
savedata "menfem.rec" /replace 
// verification 
read "menfem.rec" /close 
 
//************************************************** 
//     Exercice 3 
//     Changement d'unité statistique (aggrégation) 
//************************************************** 
 
read "femmes2010.rec" /close 
// on aggrege suivant ménage 
aggregate idmen  /close 
// on renomme la variable 
gen nbfem=n 
drop n 
// on sauve un fichier temporaire 
savedata "nbfemtemp.rec" /replace 
// on fusionne le fichier temporaire avec le fichier menages 
read "menages2010.rec" /close 
merge idmen /file="nbfemtemp.rec" 
// distribution de nbfem parmi les ménages 
freq nbfem 
freq nbfem /m 

//************************************************* 
//       Exercice 4 
//       Recodages, création de nouvelles variables 
//       Femmes 
//************************************************* 
 
// *** A partir de femmes2010.rec*************** 
 
 cd "D:\corus\maroc\FormationRabat2010\data" 
read "femmes2010.rec" /close 
// création nouvelle variable imc et codage en classes 
gen imc=poids/(taille/100)^2 
label  imc "Indice de masse corporelle en kg/m2" 
describe imc 
// recodage imc en 4 classes 
define imc4 # 
// a ce stade elle n'a que des valeurs manquantes 
//describe imc4 
// on recode imc en 4 classes 
recode imc to imc4 lo-18.4999=1 18.50-24.9999=2 25.0-29.9999=3 30-hi=4 
label imc4 "Indice de masse corporelle en 4 classes" 
// vérifications sur la variable ainsi créée 
// effectifs des modalités 
tables imc4 /M 
// avec ce means on vérifie que les minimums et maximums des classes 
// sont bien ceux que l'on veut 
means imc imc4 
// on peut assigner des labels aux codes ainsi créés 
// (en place de ceux mis par défaut par Epidata Analysis) 
labelvalue imc4 /1="1-maigre" /2="2-normal" /3="3-surpoids" /4="4-obèse" 
tables imc4  /M 
 
// création variable date de naissance format date 
gen d datenais=dmy(jnais,mnais,anais) 
//autre possibilité 
// define datenais <dd/mm/yy> 
// let datenais=dmy(jnais,mnais,anais) 
// Calcul de l'age en années 
gen age=(datenq-datenais)/365.25 
browse idfem datenq datenais age 
describe age 
// codage en classes 
define agec4 # 
recode age to agec4 lo-29.9999=1 30-39.9999=2 40-49.9999=3 50-hi=4 
means age agec4 
label age "Age en année" 
label agec4 "Age en classes de 10 ans" 
 
// nombre de personnes par année d'enquête 
gen i anenq=year(datenq) 
freq anenq 

// distribution de scoc4 
freq scoc4 
 
// recodage scoc4 en scoc3 
define scoc3 # 
recode scoc4 to scoc3 1=1 2=2 3,4=3 
tables scoc4 scoc3 
label  scoc3 "Niveau scolaire en 3 classes" 
labelvalue scoc3 /1="analphabète" /2="primaire" /3="secondaire et +" 
freq scoc3 
 
// codage obesité 
define obesite # 
recode imc to obesite lo-29.999=0 30-hi=1 
label obesite "IMC>=30 kg/m2" 
labelvalue obesite  /1="oui" /0="non" 
tables imc4 obesite 
describe obesite 
freq obesite 
 
// indicatrices de matc3 
// ce select pour ne travailler que sur les observations ayant matc3 non 
manquant 
// sinon les observations ayant données manquantes pour matc3 sont 
incorrectement 
// classées dans les catégories 0 des indicatrices  
select matc3<>. 
// génération des indicatrices : =1 ou 0 suivant que la condition est vraie ou 
fausse 
gen i matc31=(matc3=1) 
gen i matc32=(matc3=2) 
gen i matc33=(matc3=3) 
select 
describe matc31 matc32 matc33 
freq matc3 
freq matc31 matc32 matc33 
 
// variable codant l'hypertension artérielle 
// sélection des individus avec tas et tad non manquant 
 select tas<>. AND tad <>. 
 // condition vraie => hta=1  , sinon hta=0 
 gen hta=(tas>=140 OR tad>=90) 
 select 
 // effectifs 
 freq hta 
 // vérification (des maximums) 
 means tas /by=hta 
 means tad /by=hta 
 
// sauvegarde des variables initiales + nouvelles 
savedata "femmes2010_2.rec" /replace 

//************************************************* 
//       Exercice 4 
//       Recodages, création de nouvelles variables 
//       Ménages 
//************************************************* 
**** a partir de la table des ménages ********* 
read "menages2010.rec" /close 
freq depenses /cum 
// on décide de couper à 100 et 200 
define depc3 # 
recode depenses to depc3 lo-99=1 100-199=2 200-hi=3 
freq depc3 
// verification des bornes 
means depenses depc3 
Label depenses "Depenses alimentaires en trois classes" 
 
// score d'équipement du ménage 
// si variable est 1:oui 2:non 1-variable => 1:oui 2:0 
// donc on somme des variables 0/1 : le nbre total est celui des biens possédés 
gen i confort=(2-frigo)+(2-phone)+(2-tele)+(2-video)+(2-cuisine)+(2-lavess)+(2-
clime)+(2-central)+(2-parabole)+(2-lavlinge)+(2-menage)+(2-voiture) 
describe confort 
label confort "Score de confort du ménage" 
 
//Découper la variable confort en classes en fonction de la distribution. 
bar confort 
freq confort /cum 
define confc3 # 
recode confort to confc3 0,1,2=1 3=2 4-10=3 
// vérification des classes créées 
freq confc3 
tables  confort confc3 
label confc3 "Score de confort en 3 classes" 
 
// sauvegarde de la table ménage avec les nouvelles variables 
savedata "menages2010_2.rec" /replace 
// vérification de la création de la table 
read "menages2010_2.rec" /close 
describe 
 
// fusion des tables femmes2010_2 et menages2010_2 
read "femmes2010_2.rec" /close 
merge idmen /file="menages2010_2.rec" /table 
savedata " menfem2010_2.rec" / replace 
// verification 
read " menfem2010_2.rec" /close 
describe 


